Estrutura básica de um programa *Scilab* para integração de um sistema de equações diferenciais de primeira ordem

Admitiremos que o problema em foco consista na integração da equação diferencial de segunda ordem

$$m\frac{d^2x}{dt^2}+c\frac{dx}{dt}+k_{.1}x+k_2x^3=0\ ,$$

que, em espaço de estados, se descreve como

$$\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}=\begin{bmatrix} x \\ \dot{x} \end{bmatrix}\Rightarrow\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix}=\begin{bmatrix} x_2 \\ -\frac{c}{m}x_2-\frac{k_1}{m}x_1-\frac{k_2}{m}x_1^3 \end{bmatrix}$$

Um programa *Scilab* dedicado à integração desse sistema de equações diferenciais de primeira ordem deverá apresentar quatro sessões essenciais:

[1] Definição dos valores dos parâmetros das equações e demais constantes físicas

```
c=0.6
m=1;
k1=10;
k2=1;
```

[2] Definição dos estados iniciais das variáveis características do espaço de estados

```
x0=1;
xp0=0.5;
```

[3] Integração das equações diferenciais

```
x    = zeros(1000);
xp   = zeros(1000);
funcprot(0);
   function dy=massamolamortecedor(t, y)
       dy(1)=y(2);
       dy(2)=-c/m*y(2)-k1/m*y(1)-k2/m*(y(1))^3;
   endfunction;
t=linspace(0,10,1000);  // 10 segundos de simulação
y = ode([x0;xp0],0,t, massamolamortecedor);
x   = y(1,:);
xp  = y(2,:);
```

[4] Apresentação gráfica dos resultados da simulação

```
f1=scf(1);
plot(t,x);
xtitle("x em função do tempo", "t","x");
f2=scf(2);
plot(t,xp);
xtitle("dx/dt em função do tempo", "t","dx/dt");
```